

## Homework 3 Solutions

A study was conducted to investigate two treatments for patients suffering from multiple sclerosis. 150 sufferers of the disease were recruited into the study, and 75 were randomized to receive azathioprine (AZ) alone (group 1), and 75 were randomized to receive azathioprine plus methylprednisolone (AZ+MP, group 2). For each participant, a measure of auto-immunity, azathioprine APCR, was planned at clinic visits at baseline (time 0, at initiation of treatment) and at 3, 6, 9, 12, 15, and 18 months thereafter. Multiple sclerosis (MS) affects the immune system. Low values of APCR (approaching 0) are evidence that immunity is improving, which is hopefully associated with a better prognosis for sufferers of MS. Also recorded for each subject was age at entry into the study and an indicator of whether or not the subject had had previous treatment with either of the study agents (0=no, 1=yes). The average age of the men across both treatment groups was 50.45, with SD 6.69. The primary scientific aims of the study are to investigate whether (i) both treatments (AZ or AZ+MP) lower APCR over the 18 month period and (ii) whether treatment with AZ+MP results in different immune system response than does AZ alone and, if so, how it is different in terms of response over time. It was also suspected that a subject's age and prior history might be related to their APCR level at baseline and to the rate at which APCR changes during the 18 month period. The square root of APCR is the response variable of interest (square roots were taken so that the APCR observations better satisfy the assumption of normality).

### 1. (20 points) Discuss the issues of study design and how they would relate to your decision to use a linear mixed model or a pattern mixture model.

This study is a randomized longitudinal trial with repeated APCR measurements on the same subjects over time. That makes the observations within a subject correlated, so a linear mixed model is a natural primary analysis tool. It allows us to model subject-specific trajectories over the 18-month follow-up and use all available measurements, even if some participants miss visits.

A pattern mixture model could be used since the data are designed balanced. However, there are 6 measurements, which is a lot for a pattern mixture model. That is, if we did a heterogeneous variance that would be 6 variance parameters. Further, if we did an unstructured correlation that would have 15 correlation parameters. This is a lot for only 150 subjects.

So, I would go with the linear mixed model.

**2. (80 points) Regardless of your answer to (1), complete an analysis of this data using a linear mixed model. This analysis should yield the best possible answer to the study's scientific aims.**

This analysis should include:

- a. Exploratory data analysis (e.g., figures) and discussion of findings.
- b. How do we test the hypothesis of interest? Will you include any confounders or effect modifiers?
- c. Model fit: compare different mean and random effect structures.
- d. Model diagnostics checks.

## Part a

The first thing I'd like to do is to look at a plot of the data by group. I'm also going to treat group as a class variable. In SAS, I will use PROC IMPORT to read the CSV file and PROC SG PANEL / PROC SG PLOT for the plots.

```
proc import datafile="afcr.csv" out=afcr dbms=csv replace;
    guessingrows=max;
run;
```

```
proc format;
    value grpfmt 1 = "AZ" 2 = "AZ+MP";
    value prfmt 0 = "No" 1 = "Yes";
run;
```

```
proc sort data=afcr;
    by group id time;
run;
```

```
/* Raw spaghetti plot by group */
proc sgpanel data=afcr;
    panelby group / columns=2;
    series x=time y=afcr / group=id transparency=0.7;
    colaxis label="Time (months)";
    rowaxis label="sqrt(AFCR)";
run;
```

```
/* Group means on the same plot */
proc means data=afcr noprint nway;
    class group time;
    var afcr;
    output out=means_afcr mean=mean_afcr;
run;
```


```
data plot_afcr;
```


```

    set afcr(in=a) means_afcr(in=b);
run;

proc sgplot data=plot_afcr;
    series x=time y=afcr / group=id transparency=0.8;
    series x=time y=mean_afcr / group=group lineattrs=(thickness=2);
    xaxis label="Time (months)";
    yaxis label="sqrt(AFCR)";
run;

```





There does not appear to be much of an effect of the group variable. It appears that a linear time trend may be sufficient for this data. Also, there may be more heterogeneity in the later months of the study so a random slope may be required.

### Checking baseline covariates


Now let's look at whether age or prior have an impact on the baseline values of the outcome.

```
proc sgplot data=afcr;  
  where time = 0;  
  scatter x=age1 y=afcr;
```

```

loess x=age1 y=afcr / lineattrs=(thickness=2);
xaxis label="Age at entry";
yaxis label="sqrt(AFCR)";
run;

```



Clearly there is a relationship between age and AFCR. The relationship appears to be pretty linear.

Now let's check the relationship between prior:

```

proc ttest data=afcr;
  where time = 0;
  class prior;
  var afcr;
run;

```

| prior | Method | Mean    | 95% CL Mean |         | Std Dev | 95% CL Std Dev |        |
|-------|--------|---------|-------------|---------|---------|----------------|--------|
| 0     |        | 12.8064 | 12.1706     | 13.4422 | 2.1654  | 1.7994         | 2.7197 |
| 1     |        | 13.7269 | 13.1445     | 14.3093 | 2.3876  | 2.0407         | 2.8778 |

| prior         | Method        | Mean    | 95% CL Mean |         | Std Dev | 95% CL Std Dev |        |
|---------------|---------------|---------|-------------|---------|---------|----------------|--------|
| Diff (1-2)    | Pooled        | -0.9205 | -1.7872     | -0.0538 | 2.2990  | 2.0333         | 2.6451 |
| Diff (1-2)    | Satterthwaite | -0.9205 | -1.7730     | -0.0680 |         |                |        |
| Method        | Variances     | DF      | t Value     | Pr >  t |         |                |        |
| Pooled        | Equal         | 112     | -2.10       | 0.0376  |         |                |        |
| Satterthwaite | Unequal       | 104.8   | -2.14       | 0.0346  |         |                |        |

Again, a clear relationship here.

Let's see if there are relationships between age, prior, and treatment.

```
proc ttest data=afcr;
  where time = 0;
  class group;
  var age1;
run;
```

#### The TTEST Procedure

Variable: age1

| group      | Method        | N       | Mean        | Std Dev | Std Err | Minimum        | Maximum |
|------------|---------------|---------|-------------|---------|---------|----------------|---------|
| 1          |               | 55      | 50.5091     | 6.9118  | 0.9320  | 35.0000        | 65.0000 |
| 2          |               | 59      | 50.6441     | 6.6508  | 0.8659  | 32.0000        | 67.0000 |
| Diff (1-2) | Pooled        |         | -0.1350     | 6.7779  | 1.2704  |                |         |
| Diff (1-2) | Satterthwaite |         | -0.1350     |         | 1.2721  |                |         |
| group      | Method        | Mean    | 95% CL Mean |         | Std Dev | 95% CL Std Dev |         |
| 1          |               | 50.5091 | 48.6406     | 52.3776 | 6.9118  | 5.8188         | 8.5143  |
| 2          |               | 50.6441 | 48.9109     | 52.3773 | 6.6508  | 5.6301         | 8.1270  |
| Diff (1-2) | Pooled        | -0.1350 | -2.6521     | 2.3822  | 6.7779  | 5.9946         | 7.7985  |
| Diff (1-2) | Satterthwaite | -0.1350 | -2.6559     | 2.3859  |         |                |         |

| Method        | Variances | DF     | t Value | Pr >  t |
|---------------|-----------|--------|---------|---------|
| Pooled        | Equal     | 112    | -0.11   | 0.9156  |
| Satterthwaite | Unequal   | 110.68 | -0.11   | 0.9157  |

```
proc freq data=afcr;
  where time = 0;
  tables group*prior / chisq;
run;
```

The FREQ Procedure

| Frequency<br>Percent<br>Row Pct<br>Col Pct | Table of group by prior |       |       |        |
|--------------------------------------------|-------------------------|-------|-------|--------|
|                                            | group                   | prior |       |        |
|                                            |                         | 0     | 1     | Total  |
|                                            | 1                       | 19    | 36    | 55     |
|                                            |                         | 16.67 | 31.58 | 48.25  |
|                                            |                         | 34.55 | 65.45 |        |
|                                            |                         | 40.43 | 53.73 |        |
|                                            | 2                       | 28    | 31    | 59     |
|                                            |                         | 24.56 | 27.19 | 51.75  |
|                                            |                         | 47.46 | 52.54 |        |
|                                            |                         | 59.57 | 46.27 |        |
|                                            | Total                   | 47    | 67    | 114    |
|                                            |                         | 41.23 | 58.77 | 100.00 |

Statistics for Table of group by prior

| Statistic                   | DF | Value  | Prob   |
|-----------------------------|----|--------|--------|
| Chi-Square                  | 1  | 1.9586 | 0.1617 |
| Likelihood Ratio Chi-Square | 1  | 1.9672 | 0.1607 |

| Statistic                  | DF     | Value   | Prob   |
|----------------------------|--------|---------|--------|
| Continuity Adj. Chi-Square | 1      | 1.4620  | 0.2266 |
| Mantel-Haenszel Chi-Square | 1      | 1.9414  | 0.1635 |
| Phi Coefficient            |        | -0.1311 |        |
| Contingency Coefficient    |        | 0.1300  |        |
| Cramer's V                 |        | -0.1311 |        |
| <b>Fisher's Exact Test</b> |        |         |        |
| Cell (1,1) Frequency (F)   | 19     |         |        |
| Left-sided Pr <= F         | 0.1132 |         |        |
| Right-sided Pr >= F        | 0.9444 |         |        |
|                            |        |         |        |
| Table Probability (P)      | 0.0576 |         |        |
| Two-sided Pr <= P          | 0.1857 |         |        |

No relationships. However, even in a randomized trial, it is reasonable to adjust for baseline covariates such as age and prior treatment when they are strongly associated with the outcome. Randomization ensures that treatment assignment is not confounded, so adjustment is not needed to remove bias. However, including prognostic covariates can improve precision by reducing residual variability and accounting for any chance imbalances between groups. In this setting, since age and prior treatment are clearly related to AFCR, adjusting for them will lead to more efficient estimates of the treatment effects without changing their interpretation.

## Part b

The main scientific question is whether the two treatment groups differ in how AFCR changes over time. The primary test is therefore the treatment x time interaction. The model should have time, treatment, and treatment x time.

We do not need to adjust for confounders because the study is randomized. So age and prior treatment are not needed to control confounding of treatment assignment. However,



they are reasonable to include as baseline prognostic covariates because they may improve precision and account for chance imbalance.

If we thought there were some effect modifiers, we would put them in using interaction terms. Here, if there is scientific reason to believe age or prior affect the rate of change over time, then they can also be included as effect modifiers through interactions with time, treatment, or treatment x time.

## Part c

First, let's check whether a random slope is needed:

```
proc mixed data=afcr;  
  class id group prior;  
  model afcr = age1 prior group time group*time / solution ddfm=bw;  
  random intercept / subject=id;  
run;
```

| Fit Statistics           |        |
|--------------------------|--------|
| -2 Res Log Likelihood    | 3282.3 |
| AIC (Smaller is Better)  | 3286.3 |
| AICC (Smaller is Better) | 3286.3 |
| BIC (Smaller is Better)  | 3292.2 |

```
proc mixed data=afcr;  
  class id group prior;  
  model afcr = age1 prior group time group*time / solution ddfm=bw;  
  random intercept time / subject=id type=un;  
run;
```

| Fit Statistics           |        |
|--------------------------|--------|
| -2 Res Log Likelihood    | 3281.9 |
| AIC (Smaller is Better)  | 3289.9 |
| AICC (Smaller is Better) | 3289.9 |
| BIC (Smaller is Better)  | 3301.7 |

Based on this analysis, a random slope is not required (an LRT was insignificant). So, we'll use the model with just the random intercept.

Before checking some different mean specifications, we'll refit the first model using method = ml.

```
proc mixed data=afcr method=ml;
  class id group prior;
  model afcr = age1 prior group time group*time / solution ddfm=bw;
  random intercept / subject=id;
run;
```

| Fit Statistics                  |        |
|---------------------------------|--------|
| <b>-2 Log Likelihood</b>        | 3257.1 |
| <b>AIC (Smaller is Better)</b>  | 3273.1 |
| <b>AICC (Smaller is Better)</b> | 3273.3 |
| <b>BIC (Smaller is Better)</b>  | 3296.6 |

While the linear trend in AFCD by time was apparent, we'll check a profile analysis just to be complete:

```
proc mixed data=afcr method=ml;
  class id group prior timecat;
  model afcr = age1 prior group timecat group*timecat / solution ddfm=bw;
  random intercept / subject=id;
run;
```

| Fit Statistics                  |        |
|---------------------------------|--------|
| <b>-2 Log Likelihood</b>        | 3251.9 |
| <b>AIC (Smaller is Better)</b>  | 3287.9 |
| <b>AICC (Smaller is Better)</b> | 3288.8 |
| <b>BIC (Smaller is Better)</b>  | 3340.9 |

So, the linear model looks to be the best.

### Checking for effect modification (optional)

Now, let's check whether effect modification by age or prior is present. First, I'll check this out for age. I'm only going to consider a three-way age1 x group x time interaction since that is the only one that would substantially change our conclusions.

```
proc mixed data=afcr method=ml;
  class id group prior;
  model afcr = age1 prior group time group*time
              age1*time age1*group age1*group*time / solution ddfm=bw;
  random intercept / subject=id;
run;
```

| Fit Statistics                  |        |
|---------------------------------|--------|
| <b>-2 Log Likelihood</b>        | 3255.4 |
| <b>AIC (Smaller is Better)</b>  | 3277.4 |
| <b>AICC (Smaller is Better)</b> | 3277.7 |
| <b>BIC (Smaller is Better)</b>  | 3309.7 |

No effect modification for age. Now, let's check for prior:

```
proc mixed data=afcr method=ml;
  class id group prior;
  model afcr = age1 prior group time group*time
              prior*time prior*group prior*group*time / solution ddfm=bw;
  random intercept / subject=id;
run;
```

| Fit Statistics                  |        |
|---------------------------------|--------|
| <b>-2 Log Likelihood</b>        | 3252.7 |
| <b>AIC (Smaller is Better)</b>  | 3274.7 |
| <b>AICC (Smaller is Better)</b> | 3275.0 |
| <b>BIC (Smaller is Better)</b>  | 3307.1 |

No effect modification for prior. However, it is close and there is a significant interaction. So, if that was your final model then that is OK. Otherwise, we'll go back to the first model we fitted.

## Part d

Here's a summary of the final model:

```
proc mixed data=afcr method=reml;
  class id group prior;
  model afcr = age1 prior group time group*time / solution ddfm=bw;
  random intercept / subject=id;
run;
```

| Solution for Fixed Effects |       |       |          |                |     |         |         |
|----------------------------|-------|-------|----------|----------------|-----|---------|---------|
| Effect                     | group | prior | Estimate | Standard Error | DF  | t Value | Pr >  t |
| Intercept                  |       |       | 5.5772   | 0.7907         | 136 | 7.05    | <.0001  |
| age1                       |       |       | 0.1620   | 0.01491        | 648 | 10.87   | <.0001  |
| prior                      |       | 0     | -0.9569  | 0.2061         | 648 | -4.64   | <.0001  |
| prior                      |       | 1     | 0        | .              | .   | .       | .       |
| group                      | 1     |       | -0.3266  | 0.2772         | 648 | -1.18   | 0.2392  |
| group                      | 2     |       | 0        | .              | .   | .       | .       |
| time                       |       |       | -0.2473  | 0.01489        | 648 | -16.60  | <.0001  |
| time*group                 | 1     |       | 0.08645  | 0.02112        | 648 | 4.09    | <.0001  |
| time*group                 | 2     |       | 0        | .              | .   | .       | .       |

| Type 3 Tests of Fixed Effects |        |        |         |        |
|-------------------------------|--------|--------|---------|--------|
| Effect                        | Num DF | Den DF | F Value | Pr > F |
| age1                          | 1      | 648    | 118.05  | <.0001 |
| prior                         | 1      | 648    | 21.56   | <.0001 |
| group                         | 1      | 648    | 1.39    | 0.2392 |
| time                          | 1      | 648    | 373.78  | <.0001 |
| time*group                    | 1      | 648    | 16.76   | <.0001 |

Despite the first figure appearing that there was not an impact of the drug, the final model does show that the treatment was effective. Specifically, the AZ + MP group had their azathioprine decrease more rapidly than the group that had AZ alone. In the AZ + MP group we expect that the outcome ( $\sqrt{\text{AFCR}}$ ) will decrease 0.2473 units for every month increase. In the AZ group, the outcome will decrease 0.089 units less than in the AZ + MP group.